# $Id: PKGBUILD 79082 2012-10-30 09:54:22Z spupykin $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>

pkgname=miredo
pkgver=1.2.5
pkgrel=3
pkgdesc="Teredo client and server."
arch=('i686' 'x86_64')
url="http://www.remlab.net/miredo/"
license=('GPL')
depends=('judy' 'iproute2' 'libcap')
backup=('etc/miredo/miredo.conf'
        'etc/miredo/client-hook')
options=('!libtool')
source=(http://www.remlab.net/files/${pkgname}/${pkgname}-${pkgver}.tar.xz
	isatapd.rc.d
	miredo.install
	miredo.rc.d
	miredo-server.rc.d
	miredo.service)
md5sums=('5114debbf9fcab5d292176e4548f8cd1'
         '51ab6d091192605ee9206944869cb2ab'
         'd1b655d7a851cdb46c91c3418ed1962f'
         'c5a9be5c3175fecec387f1710bfd2788'
         '319aba1ae06349b76cb25fda0dba60a9'
         'abd7d8831cab7f3ea6d0d4e51b8f8f6a')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  ./configure \
	--prefix=/usr \
	--sysconfdir=/etc \
	--localstatedir=/var \
	--libexecdir=/usr/lib \
	--with-Judy
  make
  make DESTDIR="$pkgdir" install

  install -D -m755 $srcdir/miredo.rc.d $pkgdir/etc/rc.d/miredo
  install -D -m755 $srcdir/miredo-server.rc.d $pkgdir/etc/rc.d/miredo-server
  install -D -m755 $srcdir/isatapd.rc.d $pkgdir/etc/rc.d/isatapd
  sed -i 's#/sbin/ip#/usr/sbin/ip#' $pkgdir/etc/miredo/client-hook

  # avoid conflict with filesystem>=2012.06
  rmdir "$pkgdir/var/run" "$pkgdir/var"

  install -Dm644 "$srcdir/miredo.service" "$pkgdir/usr/lib/systemd/system/miredo.service"
}
